

Example 6.7 Evaluate the integral $\frac{x^2 + x + 1}{\sqrt{x^2 + 4}} dx$ using Maple's integration command. Verify the answer using our so-called elementary integral approach.

> **f:=(x^2+x+1)/sqrt(x^2+4);**

$$f := \frac{x^2 + x + 1}{\sqrt{x^2 + 4}}$$

> **ma:=int(f,x);** (ma = Maple's answer)

$$ma := \frac{1}{2}x\sqrt{x^2 + 4} - \operatorname{arcsinh} \frac{1}{2}x + \sqrt{x^2 + 4}$$

> **p:=Int(f,x);**

$$p := \frac{x^2 + x + 1}{\sqrt{x^2 + 4}} dx$$

> **with(student);**

[D, Diff, Doubleint, Int, Limit, Lineint, Product, Sum, Tripleint, changevar, completesquare, distance, equate, integrand, intercept, intparts, leftbox, leftsum, makeproc, middlebox, middlesum, midpoint, powsubs, rightbox, rightsum, showtangent, simpson, slope, summand, trapezoid]

> **p1:=expand(p);**

$$p1 := \frac{x^2}{\sqrt{x^2 + 4}} + \frac{x}{\sqrt{x^2 + 4}} + \frac{1}{\sqrt{x^2 + 4}} dx$$

Maple does not want to write this as a sum of integrals, so we separate them ourselves.

> **f1:=integrand(p1);**

$$f1 := \frac{x^2}{\sqrt{x^2 + 4}} + \frac{x}{\sqrt{x^2 + 4}} + \frac{1}{\sqrt{x^2 + 4}}$$

> **pa:=Int(op(1,f1),x);pb:=Int(op(2,f1),x);pc:=Int(op(3,f1),x);**

$$pa := \frac{x^2}{\sqrt{x^2 + 4}} dx$$

$$pb := \frac{x}{\sqrt{x^2 + 4}} dx$$

$$pc := \frac{1}{\sqrt{x^2 + 4}} dx$$

The third integral is already an elementary integral.

> **pcAns:=value(pc);**

$$pcAns := \operatorname{arcsinh} \frac{1}{2} x$$

The middle integral is an easy integration by substitution.

```
> pb1:=changevar(u=x^2+4,pb,u);
```

$$pb1 := \frac{1}{2} \frac{1}{\sqrt{u}} du$$

This is an elementary integral, so we evaluate it.

```
> pb2:=value(pb1);
```

$$pb2 := \sqrt{u}$$

```
> pbAns:=subs(u=x^2+4,pb2);
```

$$pbAns := \sqrt{x^2 + 4}$$

The first integral is more complicated. The standard approach is to turn it into a trigonometric integral with a trigonometric substitution. The term $x^2 + 4$ can be turned into a perfect square by letting $x = 2 \tan(u)$. This allows us to get rid of the radical.

```
> pa1:=changevar(x=2*tan(u),pa,u);
```

$$pa1 := 8 \frac{\tan(u)^2 (1 + \tan(u)^2)}{\sqrt{4 \tan(u)^2 + 4}} du$$

The variable u actually is defined by $u = \arctan \frac{x}{2}$. Consequently it takes its values between $-\frac{\pi}{2}$ and $\frac{\pi}{2}$. We need to tell Maple this in order to simplify the integral.

```
> assume(u>-Pi/2);additionally(u<Pi/2);
```

```
> pa2:=simplify(pa1);
```

$$pa2 := 4 \tan(u)^2 \sqrt{1 + \tan(u)^2} du$$

```
> pa3:=simplify(pa2,trig);
```

$$pa3 := 4 \tan(u)^2 \sqrt{1 + \tan(u)^2} du$$

```
> pa3:=simplify(pa2,sqrt);
```

$$pa3 := 4 \tan(u)^2 \sqrt{1 + \tan(u)^2} du$$

The whole reason for this trigonometric substitution is to get rid of the radical. We could probably find some neat Maple way of doing this, but sometimes it is easier to take stronger control over what we want to do. The next step is not elegant, but it does the job.

```
> pa3:=subs(sqrt(1+tan(u)^2)=sec(u),pa2);
```

$$pa3 := 4 \tan(u)^2 \sec(u) du$$

Let's integrate by parts choosing $dv = \sec(u) \tan(u)$ (because its an easy term to integrate).

```
> pa4:=intparts(pa3,tan(u));
```

$$pa4 := 4 \tan(u) \sec(u) - 4 (1 + \tan(u)^2) \sec(u) du$$

> **pa5:=expand(pa4);**

$$pa5 := 4 \tan(u) \sec(u) - 4 \sec(u) du - 4 \tan(u)^2 \sec(u) du$$

Notice that the integral $pa5$ has reappeared (even the multiplier of 4 has reappeared). If we let the unassigned letter w represent the value of $pa5$, we can write the equation.

> **eq:=w=op(1,pa5)+op(2,pa5)-w;**

$$eq := w = 4 \tan(u) \sec(u) - 4 \sec(u) du - w$$

> **pa6:=solve(eq,w);**

$$pa6 := 2 \tan(u) \sec(u) - 2 \sec(u) du$$

This is an elementary integral, so according to the rules of our game we are allowed to use the `value()` command.

> **pa7:=value(pa6);**

$$pa7 := 2 \tan(u) \sec(u) - 2 \ln(\sec(u) + \tan(u))$$

Since $x=2*\tan(u)$ it follows that $u=\arctan(x/2)$;

> **pa8:=subs(u=arctan(x/2),pa7);**

$$pa8 := 2 \tan \arctan \frac{1}{2} x \sec \arctan \frac{1}{2} x - 2 \ln \sec \arctan \frac{1}{2} x + \tan \arctan \frac{1}{2} x$$

> **paAns:=simplify(pa8);**

$$paAns := \frac{1}{2} x \sqrt{x^2 + 4} + 2 \ln(2) - 2 \ln(\sqrt{x^2 + 4} + x)$$

> **MyAns:=paAns+pbAns+pcAns;**

$$MyAns := \frac{1}{2} x \sqrt{x^2 + 4} + 2 \ln(2) - 2 \ln(\sqrt{x^2 + 4} + x) + \sqrt{x^2 + 4} + \operatorname{arcsinh} \frac{1}{2} x$$

> **ma;**

$$\frac{1}{2} x \sqrt{x^2 + 4} - \operatorname{arcsinh} \frac{1}{2} x + \sqrt{x^2 + 4}$$

Our answer and Maple's look quite different. Actually, all of the inverse hyperbolic functions can be expressed in terms of logarithms, but unless we take advantage of this idea, it is hard to get Maple to resolve the differences. One way to proceed is to form **ma-MyAns**, which should be zero, and ask Maple to simplify this expression. Maple usually agrees that 0 is the simplest way to express such an entry if it is zero, but in this case, Maple fails to produce the desired result. Is Maple's answer correct? Certainly, our answer is correct (we say with a smile). Let's differentiate both $MyAns$ and ma to see if we can recover $f(x)$.

> **simplify(ma-MyAns);**

$$-2 \operatorname{arcsinh} \frac{1}{2} x - 2 \ln(2) + 2 \ln(\sqrt{x^2 + 4} + x)$$

> **g:=diff(MyAns,x);**

$$g := \frac{1}{2} \sqrt{x^2 + 4} + \frac{\frac{1}{2} x^2}{\sqrt{x^2 + 4}} - \frac{2 \frac{x}{\sqrt{x^2 + 4}} + 1}{\sqrt{x^2 + 4} + x} + \frac{x}{\sqrt{x^2 + 4}} + \frac{1}{\sqrt{x^2 + 4}}$$

> **g1:=simplify(g);**

$$gl := \frac{x^2 + x + 1}{\sqrt{x^2 + 4}}$$

```
> h:=diff(ma,x);
```

$$h := \frac{1}{2}\sqrt{x^2 + 4} + \frac{\frac{1}{2}x^2}{\sqrt{x^2 + 4}} - \frac{1}{\sqrt{x^2 + 4}} + \frac{x}{\sqrt{x^2 + 4}}$$

```
> h1:=simplify(h);
```

$$h1 := \frac{x^2 + x + 1}{\sqrt{x^2 + 4}}$$

```
>
```